
How To Find A Levi Decomposition of a Lie Algebra

Synopsis
If g is a finite-dimensional Lie algebra, then the radical of g is the largest solvable 
subalgebra r contained in g . A fundamental theorem in the structure theory of Lie 
algebras asserts that there exists a complementary subalgebra s such that g is a semi-
direct sum g = s6r. This decomposition is called a Levi decomposition. The subalgebra s 
is semi-simple.
In this worksheet we show how to compute the Levi decomposition of a Lie algebra using 
the command LeviDecomposition.

Example

Load in the required packages.

with(DifferentialGeometry): with(LieAlgebras):

We shall calculate the Levi decomposition for a 7-dimensional Lie algebra g. The structure 
equations of g are 

StructureEquations := [[x1, x2] = 2*x2, [x1, x3] = 2*x3, [x1, x6] =
-2*x6, [x1, x7] = -2*x7, [x2, x4] = 2*x3, [x2, x6] = x1, [x2, x7] =
x4, [x3, x5] = x3, [x3, x6] = -x4, [x4, x5] = x4, [x4, x6] = -2*x7,
[x5, x7] = -x7];

StructureEquations d x1, x2 = 2 x2, x1, x3 = 2 x3, x1, x6 = K2 x6, x1, x7 = K2 x7,
x2, x4 = 2 x3, x2, x6 = x1, x2, x7 = x4, x3, x5 = x3, x3, x6 = Kx4, x4, x5 = x4,
x4, x6 = K2 x7, x5, x7 = Kx7

Initialize this Lie algebra with the LieAlgebraData and DGsetup commands.

LD := LieAlgebraData(StructureEquations, [x1, x2, x3, x4, x5, x6, 
x7], alg1):
DGsetup(LD);

Lie algebra: alg1

Calculate the Levi decomposition. In the output, the first list of vectors defines the radical and 
the second list of vectors defines the semi-simple subalgebra.

Levi := LeviDecomposition();
Levi d e3, e4, e5, e7 , e1, e2, e6

We can use the Query command to verify that the span of the first set of vectors is a solvable 
ideal and that the span of the second list is a semi-simple algebra.

Query(Levi[1], "Solvable");
true

Query(Levi[1], "Ideal");
true

Query(Levi[2], "Semisimple");
true

Remark. The structure of the Lie algebra becomes more transparent if we make a change of 
basis adapted to the Levi decomposition.

NewBasis := [e1, e2, e6, e5, e7, e4, e3];
NewBasis d e1, e2, e6, e5, e7, e4, e3

Calculate the structure equations of the Lie algebra in the new basis.

NewLD := LieAlgebraData(NewBasis, alg2):

Initialize the Lie algebra in the new basis. The new basis elements are labeled by X (with 
dual 1-forms labeled by omega).

DGsetup(NewLD, [X], [omega]);
Lie algebra: alg2

MultiplicationTable("LieTable");
| X1 X2 X3 X4 X5 X6 X7

---- ---- ---- ---- ---- ---- ---- ----

X1 | 0 2 X2 K2 X3 0 K2 X5 0 2 X7

X2 | K2 X2 0 X1 0 X6 2 X7 0

X3 | 2 X3 KX1 0 0 0 2 X5 X6

X4 | 0 0 0 0 KX5 KX6 KX7

X5 | 2 X5 KX6 0 X5 0 0 0

X6 | 0 K2 X7 K2 X5 X6 0 0 0

X7 | K2 X7 0 KX6 X7 0 0 0

We see that the semi-simple part of the Levi decomposition, spanned by X1, X2, X3 , is 
sl 2  with Cartan subalgebra defined by X1. We see that the radical, X4, X5, X6, X7 , 
decomposes into a 1-dimensional invariant subalgebra and a 3-dimensional irreducible 
subalgebra with highest weight vector X7. The vectors X5, X6, X7  span an abelian 
subalgebra and define the nilradical. These statements can all be verified with commands in 

the LieAlgebra package.
 
Commands Illustrated

 DGsetup, LieAlgebraData, LeviDecomposition, Query

Related Commands
Decompose, CartanSubalgebra, Nilradical, Series

Release Notes

The illustrated commands are available in Maple 11 and subsequent releases.

